THEMA

Freitag, 25. Jänner 2008
07:49



Dieses Seminar wurde am Montag 14. Jänner 2008 von 14:00 bis 17:00 Uhr im Computerraum CR 4 der HAK/HAS Tamsweg abgewickelt.

Verwendet wurde die Open Source Software Maxima Version 5.13

von http://maxima.sourceforge.net

Grundtechniken

Freitag, 25. Jänner 2008
07:29

(%i1) factor(a**2-b**2);

(%o1) $-(b-a)(b+a)$

Faktorenzerlegung

(%i2) expand(%);

(%o2) a^2-b^2

"Ausmultiplizieren"

(%i3) y=a*x**2+b*x+c;

(%o3) $y=a\,x^2+b\,x+c$

Das ist eine Gleichung

(%i4) f:rhs(f1);

(%o4) 0

Das ist ein Objekt

(%i5) f(x):=a*x**2+b*x+c;

(%o5) $f(x):=a\,x^2+b\,x+c$

Das ist eine Funktion

(%i6) f(3);

(%o6) $c+3\,b+9\,a$

(%i7) f(7);

(%o7) $c+7\,b+49\,a$

2D-Diagramme

Freitag, 25. Jänner 2008
07:24

Funktionen können leicht graphisch dargestellt werden

(%i8) f(x):=x^2-8*x+15;

(%o8) $f(x) := x^2 - 8\,x + 15$

(%i9) wxplot2d([f(x)], [x,-5,5])$

Output file "C:/Dokumente und Einstellungen/Administrator/maxout.png".

(%t9) 

(%i10) wxplot2d([f(x)], [x,-5,10], [y,-4,12])$

Output file "C:/Dokumente und Einstellungen/Administrator/maxout.png".

Man muss einen passenden Wertebereich vorsehen

(%t10) 

(%i11)

Gleichungen

Freitag, 25. Jänner 2008
07:40

Gleichungen auflösen ist eine Funktion aller bekannten Computer-Algebrasysteme

(%i1) 3*x+4=5;
(%o1) $3x+4=5$

(%i2) g1:3*x+4=5;
(%o2) $3x+4=5$

(%i3) g1;
(%o3) $3x+4=5$

Die Standardfunktion dafür ist solve()

(%i4) solve(g1,x);
(%o4) $[x=\frac{1}{3}]$

(%i5) g1:x+y=5;
(%o5) $y+x=5$

(%i6) g2:x-y=-1;
(%o6) $x-y=-1$

(%i7) solve([g1,g2],[x,y]);
(%o7) $[[x=2, y=3]]$

Es können auch allgemeine Gleichungen aufgelöst werden

(%i8) g:a*x**2+b*x+c=0;
(%o8) $a x^2+b x+c=0$

(%i9) solve(g,x);
(%o9) $[x=-\frac{\sqrt{b^2-4ac}+b}{2a}, x=\frac{\sqrt{b^2-4ac}-b}{2a}]$

(%i10)

Polynomgleichungen

Freitag, 25. Jänner 2008
07:39

Die häufig vorkommenden Polynomgleichungen können auch mit allroots() und realroots(() gelöst werden. Im Unterricht ist häufig die zweite Möglichkeiten von Interesse.

(%i10) g1:x**2-8*x+15=0;

(%o10) $x^2 - 8\,x + 15 = 0$

(%i11) g2:x**2-8*x+16=0;

(%o11) $x^2 - 8\,x + 16 = 0$

(%i12) g3:x**2-8*x+17=0;

(%o12) $x^2 - 8\,x + 17 = 0$

(%i13) allroots(g1);

(%o13) $[\,x = 3.0\,,\,x = 5.0\,]$

(%i14) allroots(g2);

(%o14) $[\,x = 4.0\,,\,x = 4.0\,]$

(%i15) allroots(g3);

(%o15) [x = 1.0 %i + 4.0 , x = 4.0 - 1.0 %i]

Wir bekommen auch die komplexen Lösungen

(%i16) %e**(%i*%pi);

(%o16) - 1

Eine der interessantesten Gleichungen, die es gibt

(%i17) realroots(g1);

Hier bekommt man nur die reellen Lösungen

(%o17) $[\,x = 3\,,\,x = 5\,]$

(%i18) realroots(g2);

(%o18) $[\,x = 4\,]$

(%i19) realroots(g3);

(%o19) [ ]

(%i20)

Gleichungssystem

Freitag, 25. Jänner 2008
07:35

```
(%i1) g1:x+y+z=3;
(%o1) z + y + x = 3

(%i2) g2:2*x-y+4*z=5;
(%o2) 4 z - y + 2 x = 5

(%i3) g3:-2*x+4*y-3*z=-1;
(%o3) - 3 z + 4 y - 2 x = - 1

(%i4) l:solve([g1,g2,g3],[x,y,z]);
(%o4) [ [ x = 1 , y = 1 , z = 1 ] ]
```

Die Lösung ist eine Liste

```
(%i5) l[1];
(%o5) [ x = 1 , y = 1 , z = 1 ]

(%i6) l[1][1];
(%o6) x = 1

(%i7) x:ev(x,l[1][1]);
(%o7) 1
```

Die Lösung für x freistellen

```
(%i8) x;
(%o8) 1
```

Matrizenmethode

Freitag, 25. Jänner 2008
07:34

(%i9) g1;

(%o9) $z + y + x = 3$

(%i10) g2;

(%o10) $4z - y + 2x = 5$

(%i11) g3;

(%o11) $-3z + 4y - 2x = -1$

(%i12) A:matrix(
[1,1,1],
[2,-1,4],
[-2,4,-3]
);

(%o12) $\begin{bmatrix} 1 & 1 & 1 \\ 2 & -1 & 4 \\ -2 & 4 & -3 \end{bmatrix}$

Die Eingabe der Matrix erfolgt über das Menü von wxMaxima

/*

Inverse Matrix

(%i13) B:invert(A);

(%o13) $\begin{bmatrix} \frac{13}{9} & -\frac{7}{9} & -\frac{5}{9} \\ \frac{2}{9} & \frac{1}{9} & \frac{2}{9} \\ -\frac{2}{3} & \frac{2}{3} & \frac{1}{3} \end{bmatrix}$

Berechnung der inversen Matrix

/*

der Vektor b ist eine einspaltige Matrix

(%i14) b:matrix(
[3],
[5],
[-1]
);

(%o14) $\begin{bmatrix} 3 \\ 5 \\ -1 \end{bmatrix}$

/*

Die Lösung durch Matrizenmultiplikation

(%i15) x:B.b;

(%o15) $\begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$

Das ist das Ergebnis: x=1, y=1 und z=1

Lineare Optimierung

Freitag, 25. Jänner 2008
07:23

/*

Simplexverfahren

(%i1) load(simplex);

Man muss ein Unterprogramm laden

(%o1)
C:/Programme/Maxima-5.13.0/share/maxima/5.13.0/share/contrib/simplex/simplex.mac

(%i2) u1:x>=0;

(%o2) $x \geq 0$

Nichtnegativitätsbedinungen

(%i3) u2:y>=0;

(%o3) $y \geq 0$

(%i4) u3:x<=10;

Nebenbedingungen formulieren

(%o4) $x \leq 10$

(%i5) u4:x+2*y<=12;

(%o5) $2\,y + x \leq 12$

(%i6) u5:3*x+y<=16;

(%o6) $y + 3\,x \leq 16$

(%i7) zf:12*x+14*y;

Das ist die Zielfunktion

(%o7) $14\,y + 12\,x$

(%i8) NB:[u1,u2,u3,u4,u5];

(%o8) $[\,x \geq 0\,,\; y \geq 0\,,\; x \leq 10\,,\; 2\,y + x \leq 12\,,\; y + 3\,x \leq 16\,]$

(%i9) maximize_lp(zf,NB);

Das ist die Lösung der Maximumsaufgabe

(%o9) $[\,104\,,\,[\,y = 4\,,\; x = 4\,]\,]$

Folgen

Freitag, 25. Jänner 2008
07:21

```
(%i1) a:makelist(n,n,1,7);
(%o1) [ 1 , 2 , 3 , 4 , 5 , 6 , 7 ]

(%i2) a:makelist(3*i+4,i,1,10);
(%o2) [ 7 , 10 , 13 , 16 , 19 , 22 , 25 , 28 , 31 , 34 ]

(%i3) a[3];
(%o3) 13

(%i4) a[7];
(%o4) 25
```

Folgen sind in Form von Listen das zentrale Objekt von Maxima

Summen

Freitag, 25. Jänner 2008
07:19

```
(%i5) sum(2*n-1,n,1,10);
(%o5) 100

(%i6) sum(n,n,1,inf);
```

(%o6) $\sum_{n=1}^{\infty} n$

```
(%i7) %,simpsum;
(%o7) ∞

(%i8) sum(1/2**(i-1),i,1,inf);
```

(%o8) $\sum_{i=1}^{\infty} 2^{1-i}$

```
(%i9) %,simpsum;
(%o9) 2

(%i10) sum(2**(i-1),i,1,64);
(%o10) 18446744073709551615
/*
```

Differentialquotient

Freitag, 25. Jänner 2008
07:16

/*

Grenzwerte

(%i11) k(x,h):=(f(x+h)-f(x))/h;

(%o11) $k(x,h):=\frac{f(x+h)-f(x)}{h}$

(%i12) f(x):=x**2;

(%o12) $f(x):=x^2$

(%i13) limit(k(x,h),h,0);

(%o13) $2x$

(%i14) f(x):=x**3;

(%o14) $f(x):=x^3$

(%i15) limit(k(x,h),h,0);

(%o15) $3x^2$

(%i16) f(x):=x**n;

(%o16) $f(x):=x^n$

(%i17) limit(k(x,h),h,0);

(%o17) $n\,x^{n-1}$

(%i18) f(x):=u(x)+v(x);

(%o18) $f(x):=u(x)+v(x)$

(%i19) limit(k(x,h),h,0);

(%o19) $\lim_{h \to 0} \frac{v(x+h)+u(x+h)-v(x)-u(x)}{h}$

(%i20) ab:diff(f(x),x);

(%o20) $\frac{d}{dx}v(x)+\frac{d}{dx}u(x)$

(%i21) f(x):=sin(x)*exp(x);

(%o21) $f(x):=\sin(x)\exp(x)$

(%i22) ab:diff(f(x),x);

(%o22) $\%e^x \sin(x)+\%e^x \cos(x)$

(%i23) limit(k(x,h),h,0);

Is x positive, negative, or zero? zero;

Is x an integer ? yes;

(%o23) $\%e^x(\sin(x)+\cos(x))$

Mit Hilfe der Grenzwertrechnung kann man Ableitung bestimmen

Kurvendiskussion

Freitag, 25. Jänner 2008
07:13

(%i1) f(x):=x**3-8*x**2+15*x+12;

(%o1) $f(x) := x^3 - 8x^2 + 15x + 12$

(%i2) n:realroots(f(x));

Nur reelle Lösungen sind interessant

(%o2) $[x = -\frac{20007043}{33554432}]$

(%i3) %,numer;

(%o3) $[x = -0.59625634551048]$

(%i4) solve(f(x)=0,x);

Dasselbe Problem kann auch mit solve() gelöst werden

(%o4) $[x = -\frac{19\left(\frac{\sqrt{3}\,\%i}{2}-\frac{1}{2}\right)}{3\,19^{1/3}} - \frac{19^{1/3}\left(-\frac{\sqrt{3}\,\%i}{2}-\frac{1}{2}\right)}{3} + \frac{8}{3}, x = -\frac{19^{1/3}\left(\frac{\sqrt{3}\,\%i}{2}-\frac{1}{2}\right)}{3} - \frac{19\left(-\frac{\sqrt{3}\,\%i}{2}-\frac{1}{2}\right)}{3\,19^{1/3}} + \frac{8}{3}, x = -\frac{19^{1/3}}{3} - \frac{19}{3\,19^{1/3}} + \frac{8}{3}]$

(%i5) %,numer;

(%o5) [x = - 2.373455786300665 (0.86602540378444 %i - 0.5) - 0.88946721624065 (- 0.86602540378444 %i - 0.5) + 2.666666666666667 , x = - 0.88946721624065 (0.86602540378444 %i - 0.5) - 2.373455786300665 (- 0.86602540378444 %i - 0.5) + 2.666666666666667 , x = - 0.59625633587465]

(%i6) ab:diff(f(x),x);

Mit diff() kann man Ableitungen bestimmen

(%o6) $3x^2 - 16x + 15$

(%i7) e:realroots(ab);

(%o7) $[x = \frac{40725025}{33554432}, x = \frac{138231945}{33554432}]$

(%i8) f(x),e[1],numer;

(%o8) 20.20882073535354

(%i9) f(x),e[2],numer;

(%o9) 7.93932741279461

(%i10) ab2:diff(ab,x);

(%o10) $6x - 16$

(%i11) ab2:diff(f(x),x,2);

Hier geht es um die zweite Ableitung

(%o11) $6x - 16$

Pole

Freitag, 25. Jänner 2008
07:10

(%i12) f(x):=(x**2+4*x+12)/(x**2-8*x+15);

(%o12) $f(x):=\frac{x^2+4x+12}{x^2-8x+15}$

Eine gebrochen rationale Funktion

(%i13) n:denom(f(x));

(%o13) $x^2-8x+15$

Bestimmung des Nenners

(%i14) num(f(x));

(%o14) $x^2+4x+12$

Bestimmung des Zählers

(%i15) solve(n=0,x);

(%o15) $[x=3, x=5]$

Die Pole sind die Nullstellen des Nenners

Integralrechnung

Freitag, 25. Jänner 2008
07:00

(%i17) f(x):=x**2;

(%o17) $f(x):=x^2$

(%i18) integrate(f(x),x);

(%o18) $\frac{x^3}{3}$

Unbestimmtes Integral

(%i19) integrate(f(x),x,0,1);

Bestimmtes Integral

(%o19) $\frac{1}{3}$

(%i20) 'integrate(f(x),x,0,1);

(%o20) $\int_0^1 x^2 \, dx$

(%i21) 'integrate(f(x),x,0,1)=integrate(f(x),x,0,1);

Das einfache Hochkomma verhindert die Auswertung

(%o21) $\int_0^1 x^2 \, dx = \frac{1}{3}$

Partialbruchzerlegung

Freitag, 25. Jänner 2008
06:58

(%i22) f(x):=x/(x**2-8*x+15);

(%o22) $f(x):=\frac{x}{x^2-8x+15}$

(%i23) integrate(f(x),x);

(%o23) $\frac{5\log(x-5)}{2}-\frac{3\log(x-3)}{2}$

Dafür habe ich als Schüler eine Stunde gebraucht

(%i24) partfrac(f(x),x);

(%o24) $\frac{5}{2(x-5)}-\frac{3}{2(x-3)}$

Ereignisalgebra

Freitag, 25. Jänner 2008
12:53

```
(%i1) S:{1,2,3,4,5,6};
(%o1) {1,2,3,4,5,6}
```

Ein Würfel

```
(%i2) G:subset(S,lambda([e],is(mod(e,2)=0)));
(%o2) {2,4,6}
```

Es kommt eine gerade Zahl

```
(%i4) U:setdifference(S,G);
(%o4) {1,3,5}
```

Das Komplementärereignis

```
(%i5) m:cardinality(S);
(%o5) 6
```

Die Mächtigkeit von Mengen

```
(%i6) g[G]:cardinality(G);
(%o6) 3
```

```
(%i9) W[G]:g[G]/m;
```

(%o9) $\frac{1}{2}$

Wahrscheinlichkeit = Anzahl der günstigen Fälle / Anzahl der möglichen Fälle

```
(%i11) W[U]:1-W[G];
```

(%o11) $\frac{1}{2}$

Wahrscheinlichkeit des Komplementärereignisses

```
(%i12)
```

Permutationen

Freitag, 25. Jänner 2008
16:10

```
(%i1) S:{1,2,3};
(%o1) {1,2,3}

(%i2) P:permutations(S);
(%o2) {[1,2,3],[1,3,2],[2,1,3],[2,3,1],[3,1,2],[3,2,1]}

(%i3) n:cardinality(P);
(%o3) 6

(%i4) n:cardinality(S)!;
(%o4) 6
```

Die Anzahl der möglichen Anordnung von n Elementen ist n!

Kombinationen

Freitag, 25. Jänner 2008
16:18

```
(%i5) S:{1,2,3,4,5,6};
(%o5) {1,2,3,4,5,6}

(%i6) T:powerset(S);
(%o6) {{},{1},{1,2},{1,2,3},{1,2,3,4},{1,2,3,4,5},{1,2,3,4,5
,6},{1,2,3,4,6},{1,2,3,5},{1,2,3,5,6},{1,2,3,6},{1,2,4},{1,
2,4,5},{1,2,4,5,6},{1,2,4,6},{1,2,5},{1,2,5,6},{1,2,6},{1,
3},{1,3,4},{1,3,4,5},{1,3,4,5,6},{1,3,4,6},{1,3,5},{1,3,5,
6},{1,3,6},{1,4},{1,4,5},{1,4,5,6},{1,4,6},{1,5},{1,5,6},{1
,6},{2},{2,3},{2,3,4},{2,3,4,5},{2,3,4,5,6},{2,3,4,6},{2,3,
5},{2,3,5,6},{2,3,6},{2,4},{2,4,5},{2,4,5,6},{2,4,6},{2,5},
{2,5,6},{2,6},{3},{3,4},{3,4,5},{3,4,5,6},{3,4,6},{3,5},{3,
5,6},{3,6},{4},{4,5},{4,5,6},{4,6},{5},{5,6},{6}}

(%i8) Z:subset(T,lambda([e],is(cardinality(e)=3)));
(%o8) {{1,2,3},{1,2,4},{1,2,5},{1,2,6},{1,3,4},{1,3,5},{1,3,
6},{1,4,5},{1,4,6},{1,5,6},{2,3,4},{2,3,5},{2,3,6},{2,4,5},
{2,4,6},{2,5,6},{3,4,5},{3,4,6},{3,5,6},{4,5,6}}

(%i9) b:cardinality(Z);
(%o9) 20

(%i10) b:binomial(cardinality(S),3);
(%o10) 20

(%i11)
```

Die Möglichkeit aus einer Menge mit n Elementen Teilmengen der Mächtigkeit k zu bestimmen

Binomialverteilung

Samstag, 26. Jänner 2008
17:50

wxMaxima session 1 / 1

(%i1) n:10;

(%o1) 10

(%i2) p:0.1;

(%o2) 0.1

(%i3) W(k):=binomial(n,k)*p**k*(1-p)**(n-k);

(%o3) $W(k) := \binom{n}{k} p^{k} (1-p)^{n-k}$

Die bekannte Formel für die Binomialverteilung

(%i4) P:makelist(W(k),k,0,n);

(%o4) $[\,0.3486784401,\ 0.387420489,\ 0.1937102445,\ 0.057395628,\ 0.011160261,\ 0.0014880348,\ 1.377810000000001\ 10^{-4},\ 8.7480000000000067\ 10^{-6},\ 3.6450000000000033\ 10^{-7},\ 9.0000000000000078\ 10^{-9},\ 1.0000000000000011\ 10^{-10}\,]$

(%i5) transpose(P);

Transponieren macht aus der Zeile eine Spalte (übersichtlichere Anzeige!)

(%o5) $\begin{bmatrix} 0.3486784401 \\ 0.387420489 \\ 0.1937102445 \\ 0.057395628 \\ 0.011160261 \\ 0.0014880348 \\ 1.377810000000001\ 10^{-4} \\ 8.7480000000000067\ 10^{-6} \\ 3.6450000000000033\ 10^{-7} \\ 9.0000000000000078\ 10^{-9} \\ 1.0000000000000011\ 10^{-10} \end{bmatrix}$

(%i6)

Diese Dokumentation wurde erstellt mit MS Office OneNote 2007